

How does the current date function work?


In order to fully utilize Business Object's ability to automate reports, you may need to restrict information based on the run date of a report. For example, you may have a report which runs on the 10th of every month, but needs to return data through the end of previous month. Or you may want to schedule a recurring instance for an entire year, but each instance should be able to provide a month-to-date (MTD) calculation.

By utilizing the current date function and crafting variables in your report which take advantage of the function, you can create reports in which the query date range does not need to be updated with each run or recurrence.

In this example, I will run the Budget vs Actual query through posting period 16, but will restrict calculations to return only values from the end of the previous month.

Creating Variables and using Current Date function:

For this example, the report is being run in February, period 8, and the report will be filtered through January, period 7:



1. Create a variable to retrieve the Run Date of the report.

=currentdate()

- Name this variable **Current Date**.
- This variable will return today's date in this format "M/DD/YY" (2/25/16).

2. Create another variable to extract the numerical month as a string from the current date variable.

=FormatDate([Current Date];"M")

- Name this Variable **Current Month**.
- This variable will return "2" if the current month is February.

You now have two variables. One that returns the current date and another that uses that date and returns the month.

Current Date	Current Month
2/25/16	2

3. To derive positing period based on calendar month, we can then use the Current Month variable in an If/Then statement. The If/Then statement will return a new value, the post period:

```
=if ([Current Month]="1") then 7
Elseif ([Current Month]="2") then 8
Elseif ([Current Month]="3") then 9
Elseif ([Current Month]="4") then 10
Elseif ([Current Month]="5") then 11
Elseif ([Current Month]="6") then 12
Elseif ([Current Month]="7") then 1
Elseif ([Current Month]="8") then 2
Elseif ([Current Month]="9") then 3
Elseif ([Current Month]="10") then 4
Elseif ([Current Month]="11") then 5
Elseif ([Current Month]="12") then 6
```

- Name this variable **Current Month PP**
- This variable will return "8" if the Current Month is 2.

Using Current Date in Scheduled Reports

Note: By not putting quotation marks around the resulting value, we have told Business Objects to treat these values as numbers. This is an important distinction because strings (denoted by quotation marks) cannot be tested in value comparisons such as greater than or equal to. However, numerical values can.

You should now have 3 variables:

Current Date	Current Month	Current Month PP
2/25/16	2	8

- We'll need to take the Current Month PP variable and use it to determine the previous Posting Period. We can do that with an If/Then statement. This is necessary because when we run the report in July, we'll want all previous periods for the previous fiscal year.

=If([Current Month PP]= 1) Then 12

Else [Current Month PP]-1

- Name this variable **Previous PP**.
- This formula states that if the Current PP is July (PP1), then the Previous Period would be 12 and all other previous periods would be the Current Month PP value minus 1. In February, the Current Month PP value will be 8, and this variable would return 7.

Current Date	Current Month	Current Month PP	Previous PP
2/25/16	2	8	7

- We now have to turn the Posting Period from the query into a number. Because of the “#” statistical posting period, we have to use an if/then statement to turn that period into 0 and the rest of the posting periods into numbers.

- =If([Posting period]="#") Then 0 Else ToNumber([Posting period])** Name this variable **PP**.
- This variable turns all posting periods into a number (rather than a string) and turns the # statistical posting period, #, into a “0”.

Note: Different queries may have a different format for Posting Period. For example, In the Budget vs Actual, the Posting Period is listed as “[Posting period]”, but in other queries it could be “[Posting Period]”. Be mindful of the format of the object when using it in your formula.

Here's a demonstration of the variable:

Posting period	PP
#	0
1	1
2	2
3	3
4	4
5	5
6	6
7	7
8	8
9	9
10	10
11	11
12	12
13	13
14	14
15	15
16	16


6. Once we have converted the posting period to a number, we can now create a variable to return only transactions that are less than or equal to the previous Posting Period.

=If([PP]<=[Previous PP]) Then 1 Else 0

- Name this variable **Filter**.
- You can then use a filter and inlist operator to return only transactions that have a marked "1" by the variable, thus anything that occurred in the current month would then be filtered out.

Using Current Date in Scheduled Reports

Apply the filter to the table:



Your table has now been filtered for transactions that are equal to or less than the previous posting period:

Before the filter has been applied				After the filter has been applied			
Vendor	Posting period	Filter	YTD Actual Expense	Vendor	Posting period	Filter	YTD Actual Expense
VERIZON WIRELESS	#	1		VERIZON WIRELESS	#	1	
VERIZON WIRELESS	1	1	\$546.84	VERIZON WIRELESS	1	1	\$546.84
VERIZON WIRELESS	2	1	\$7,465.95	VERIZON WIRELESS	2	1	\$7,465.95
VERIZON WIRELESS	3	1	\$35,438.72	VERIZON WIRELESS	3	1	\$35,438.72
VERIZON WIRELESS	4	1	\$13,320.22	VERIZON WIRELESS	4	1	\$13,320.22
VERIZON WIRELESS	5	1	\$12,287.56	VERIZON WIRELESS	5	1	\$12,287.56
VERIZON WIRELESS	6	1	\$27,205.11	VERIZON WIRELESS	6	1	\$27,205.11
VERIZON WIRELESS	7	1	\$15,958.73	VERIZON WIRELESS	7	1	\$15,958.73
VERIZON WIRELESS	8	0	\$7,489.85				
		Sum:	\$119,712.98			Sum:	\$112,223.13

How to create a MTD variable using the current date

MTD measures from the query are calculated based on the posting period input from the prompt. If you set the query to run for a range of time or for a period such as 16, you won't be able to use the MTD measure from the query, but you create your own variable to calculate the MTD value based on the current date.

MTD measures found in query prompts use the user input to calculate the MTD measures. However, if you're scheduling a query to run for a range of time (such as fiscal or calendar year) then these measures become irrelevant or inaccurate. This can be remedied by creating a variable to base its MTD calculation on the current run date.

Below are two examples in which you might utilize:

1. Create a Previous MTD variable: Use this formula and create your variable as a measure.

=[YTD Actual Expense] Where ([PP]=[Previous PP])

- Name this variable **MTD Previous Month**.
- This variable will select records from the query where the posting period matches the run date previous posting period variable, thus capturing expenses incurred in the previous posting period while ignoring other posting periods.

2. Create a current MTD variable.

=[YTD Actual Expense] Where ([PP] = [Current Month PP])

- Name this variable **MTD Current Month**.
- This variable will return expenses where the Posting Period is equal to Current Posting Period.

Vendor	Posting period	Filter	YTD Actual Expense	MTD Previous Month
VERIZON WIRELESS	#	1		
VERIZON WIRELESS	1	1	\$546.84	
VERIZON WIRELESS	2	1	\$7,465.95	
VERIZON WIRELESS	3	1	\$35,438.72	
VERIZON WIRELESS	4	1	\$13,320.22	
VERIZON WIRELESS	5	1	\$12,287.56	
VERIZON WIRELESS	6	1	\$27,205.11	
VERIZON WIRELESS	7	1	\$15,958.73	\$15,958.73
VERIZON WIRELESS	8	0	\$7,489.85	
		Sum:	\$119,712.93	\$15,958.73

Vendor	Posting period	Filter	YTD Actual Expense	MTD Current Month
VERIZON WIRELESS	#	1		
VERIZON WIRELESS	1	1	\$546.84	
VERIZON WIRELESS	2	1	\$7,465.95	
VERIZON WIRELESS	3	1	\$35,438.72	
VERIZON WIRELESS	4	1	\$13,320.22	
VERIZON WIRELESS	5	1	\$12,287.56	
VERIZON WIRELESS	6	1	\$27,205.11	
VERIZON WIRELESS	7	1	\$15,958.73	
VERIZON WIRELESS	8	0	\$7,489.85	\$7,489.85
		Sum:	\$119,712.93	\$7,489.85

Variable Summary

Current Date	=CurrentDate()
Current Month	=FormatDate([Current Date];"M")
Current Month PP	=If ([Current Month]="1") Then 7 Elseif ([Current Month]="2") Then 8 Elseif ([Current Month]="3") Then 9 Elseif ([Current Month]="4") Then 10 Elseif ([Current Month]="5") Then 11 Elseif ([Current Month]="6") Then 12 Elseif ([Current Month]="7") Then 1 Elseif ([Current Month]="8") Then 2 Elseif ([Current Month]="9") Then 3 Elseif ([Current Month]="10") Then 4 Elseif ([Current Month]="11") Then 5 Elseif ([Current Month]="12") Then 6
Previous PP	=If ([Current Month PP]= 1) Then 12 Else [Current Month PP]-1
PP	=If ([Posting period]="#") Then 0 Else ToNumber([Posting period])
Filter	=If ([PP]<[Current Month PP]) Then 1 Else 0
MTD Previous Month	=[YTD Actual Expense] Where ([PP]=[Previous PP])
MTD Current Month	=[YTD Actual Expense] Where ([PP] = [Current Month PP])